﻿Shader "Unity Shader Book/Chapter 6/SpecularVertexLevel"
{
    Properties
    {
        _Diffuse("Diffuse",Color) = (1,1,1,1)
        _Specular("Specular",Color) = (1,1,1,1)
        _Gloss("Gloss",Range(8.0 , 256)) = 20
    }

    SubShader
    {
        Pass {
            Tags{ "LightMode" = "ForwardBase" } //定义 该Pass 在Unity的光照流水线的角色（只有定义了正确的LightMode，才能得到一些Unity的内置光照变量，如 _LightColor0）

            CGPROGRAM

            #pragma vertex vert
            #pragma fragment frag
            #include "Lighting.cginc"

            fixed4 _Diffuse;
            fixed4 _Specular;
            float _Gloss;

            struct a2v{
                float4 vertex : POSITION;
                fixed3 normal : NORMAL;
            };

            struct v2f{
                float4 pos: SV_POSITION;
                fixed3 color: COLOR;
            };

            v2f vert(a2v v){
                v2f o;
                
                o.pos = UnityObjectToClipPos(v.vertex);

                fixed3 ambient = UNITY_LIGHTMODEL_AMBIENT.xyz;

                fixed3 worldNormal = normalize(mul(v.normal, (float3x3)unity_WorldToObject) );

                fixed3 worldLightDir = normalize (_WorldSpaceLightPos0.xyz);

                fixed3 diffuse = _LightColor0.rgb * _Diffuse.rgb * saturate(dot(worldNormal, worldLightDir));

                fixed3 reflectDir = normalize ( reflect (-worldLightDir, worldNormal));

                fixed3 viewDir = normalize(_WorldSpaceCameraPos.xyz - mul(unity_ObjectToWorld, v.vertex).xyz );

                fixed3 specular = _LightColor0.rgb * _Specular.rgb * pow ( saturate ( dot (reflectDir , viewDir )), _Gloss);

                o.color = ambient + diffuse + specular;

                return o;
            }

            fixed4 frag (v2f i) : SV_TARGET{
                return fixed4 (i.color,1.0);
            }

            ENDCG
        }
    }
    Fallback "Specular"
}
